```
1 Print "  RELIABILITY ANALYSIS OF REDUNDANT SYSTEMS (RMN.BAS)"
2 Print "  Written by Dwight S. Burgess 11-05-1986
3 Print "  Revised by Phil Rutherford 10-22-2020
4 Print "  Run with MMBasic.exe (www.mmbasic.com)"
5 Print
13 Print "  This program calculates reliability of redundant systems based on the general"
14 Print "  equation developed By Robert S.Pringle And Philip M.Gresho in"
15 Print "  'A Comprehensive Reliability Analysis Of Redundant Systems',"
16 Print "  Journal Of Spacecraft And Rockets, Vol. 4, No.5, May 1967, pp 631-638."
17 Print "  Reliability logic options include parallel operating redundancy or operating"
18 Print "  and standby redundancy with or without startup for operating units and with"
19 Print "  or without switching for employed standby units."
20 Print
22 '-----------------------------------------
23 'RELIABILITY PROGRAM FOR STANDBY REDUNDANCY
24 '-----------------------------------------
34 Print TAB(71); Date$
35 Print TAB(71); "DD-MM-YYYY"
36 Print TAB(73); TIME$
37 Print
38 Input "  WHAT IS THE OPERATING FAILURE RATE (/HR)";LAMDOP
39 Input "  WHAT IS THE STANDBY FAILURE RATE (/HR)";LAMSTAND
40 Input "  WHAT IS THE MISSION TIME (HR)";T
50 Input "  WHAT IS THE PROBABILITY OF SUCCESSFUL STARTUP FOR EACH OPERATING AND STANDBY UNIT";RSTART
60 Input "  WHAT IS THE PROBABILITY OF SUCCESSFULLY SWITCHING A STARTED STANDBY UNIT ON-LINE";RSWITCH
70 Input "  WHAT IS THE REQUIRED NUMBER OF OPERATING UNITS (M)";M
80 Input "  HOW MANY STANDBY (OR EXTRA OPERATING) UNITS ARE PROVIDED (N)";N
90 IF LAMSTAND =0 THEN BETA=1 ELSE BETA=1-EXP(-LAMSTAND*T)
100 ALPHA=M*LAMDOP/LAMSTAND
110 ROP=EXP(-M*LAMDOP*T)
120 IF M>N THEN ETA=N ELSE ETA=M
130 RSTARTSW=RSTART*RSWITCH
132 Print
135 Print "  INPUT DATA"
137 Print
140 Print "  M =";M
142 Print "  N =";N
143 Print "  LAMDOP =";LAMDOP;"(/HR)"
146 Print "  LAMSTAND =";LAMSTAND;"(/HR)"
148 Print "  MISSION TIME ="T;"(HR)"
150 Print "  RSTART ="RSTART
152 Print "  RSWITCH ="RSWITCH
155 Print "  RSTARTSW ="RSTARTSW
160 Print "  ALPHA (M*LAMDOP/LAMSTAND) =";ALPHA
162 Print "  BETA (UNRELIABILITY OF SINGLE STANDBY UNIT) ="BETA
164 Print "  ROP (RELIABILITY OF (=M/M.S) OPERATING UNITS) ="ROP
```

```
166 Print "  ETA (MINIMUM OF M AND N) ="ETA
170 Print
180 GOTO 6000
2000 I=1
2010 FOR J=1 TO NUM
2020 I=I*J:NEXT J
2030 Return
2040 '--------------------
3000 'SUBROUTINE FOR COEF1
3010 '--------------------
3020 NUM = M:GOSUB 2000
3030 MFAC = I
3040 NUM = K:GOSUB 2000
3050 KFAC = I
3060 NUM = M-K:GOSUB 2000
3070 MKFAC=I
3080 COEF1=MFAC/(KFAC*MKFAC)
3090 Return
3095 '--------------------
4000 'SUBROUTINE FOR COEF2
4005 '--------------------
4010 PROD = 1
4020 FOR L = 1 TO Y
4030 PROD=PROD*(Y-L+ALPHA)
4040 NEXT L
4050 NUM = Y:GOSUB 2000
4060 YFAC = I
4070 COEF2=PROD/YFAC
4080 Return
4090 '-------------------------------------------------
4100 'ALTERNATE SUBROUTINE FOR COEF2 WHEN LAMSTAND = 0
4101 '-------------------------------------------------
4110 NUM=Y:GOSUB 2000
4120 YFAC=I
4130 COEF2=((M*LAMDOP*T)^Y)/YFAC
4140 Return
4200 '--------------------
5000 'SUBROUTINE FOR COEF3
5001 '--------------------
5010 NUM=N:GOSUB 2000
5020 NFAC=I
5030 NUM=X:GOSUB 2000
5040 XFAC=I
5050 NUM=N-X:GOSUB 2000
5060 NXFAC=I
5070 COEF3=NFAC/(XFAC*NXFAC)
5080 Return
5500 '---------------------------------
6000 'COMBINATORIAL EQUATIONS START HERE
6001 '---------------------------------
6010 RMN=0:SUM1=0
```

```
 98 6020 FOR K=0 TO ETA
 99 6030 SUM2=0
100 6040 FOR Y=0 TO N-K
101 6050 SUM3=0
102 6060 FOR X=0 TO N-K-Y
103 6070 GOSUB 5000
104 6080 SUM3=SUM3+COEF3*(1-RSTARTSW)^X*RSTARTSW^(N-X)
105 6090 NEXT X
106 6100 IF LAMSTAND=0 THEN GOSUB 4100 ELSE GOSUB 4000
107 6110 SUM2=SUM2+COEF2*BETA^Y*ROP*SUM3
108 6120 NEXT Y
109 6130 GOSUB 3000
110 6140 SUM1=SUM1+COEF1*(1-RSTART)^K*RSTART^(M-K)*SUM2
111 6150 NEXT K
112 6160 RMN=SUM1
113 6170 Print "  SYSTEM RELIABILITY IS";RMN;"  P(=>"M"/"M+N".S)"
114 6180 '--------------
115 6200 'PRINTED OUTPUT
116 6201 '--------------
117 6230 Print
118 6250 Print
119 6300 Print "                                                                                                    ↵
    ____________________________________________________________________________________________________"
120 6400 Print
121 6450 Print TAB(71); Date$
122 6455 Print TAB(71); "DD-MM-YYYY"
123 6460 Print TAB(73); TIME$
124 6500 Print
125 7000 Print"  CALCULATED RELIABILITY FOR A REDUNDANT SYSTEM WITH THE FOLLOWING                              ↵
    CONFIGURATION:
126 7010 Print
127 7020 IF LAMDOP<>LAMSTAND THEN 7080
128 7030 Print "  PARALLEL OPERATING REDUNDANCY WITH";M;" UNITS REQUIRED OUT OF";M+N;"                          ↵
    OPERATING UNITS"
129 7035 Print
130 7040 Print TAB(5)"MISSION TIME IS";TAB(50)T;" HR"
131 7050 Print TAB(5)"OPERATING FAILURE RATE IS";TAB(50)LAMDOP;" /HR"
132 7060 Print TAB(5)"UNIT STARTUP RELIABILITY IS";TAB(50)RSTART
133 7070 Print TAB(5)"SYSTEM RELIABILITY IS P(=>"M"/"M+N".S)";TAB(50)RMN:Goto 7200
134 7080 Print "  STANDBY REDUNDANCY WITH";M;" OPERATING UNITS REQUIRED";
135 7090 Print " AND";N;" STANDBY UNITS"
136 7095 Print TAB(5)"MISSION TIME IS";TAB(50)T;"HR"
137 7100 Print TAB(5)"OPERATING UNIT FAILURE RATE IS";TAB(50)LAMDOP;" /HR"
138 7110 Print TAB(5)"STANDBY UNIT FAILURE RATE IS";TAB(50)LAMSTAND;" /HR"
139 7130 Print TAB(5)"PROBABILITY OF UNIT STARTUP IS";TAB(50)RSTART
140 7160 Print TAB(5)"PROBABILITY OF SWITCHING A STANDBY ON-LINE IS";TAB(50)RSWITCH
141 7180 Print TAB(5)"SYSTEM RELIABILITY IS P(=>"M"/"M+N".S)";TAB(50)RMN
142 7200 Print"                                                                                                    ↵
    ____________________________________________________________________________________________________"
143 7210 Print
144 8000 END
```

145